

What?

Investors care about financial ratios. However, implications and importance of financial ratios may be dependent on several factors. Therefore, market's perception of each financial ratio changes through time.

We would expect that the companies that have the highest gross margins would be favorable to investors at all times. However, our job is to figure out the current investor appetite for the gross margin.

Method

We will create two portfolios based on the gross margin. *Portfolio_{HIGH}* will include companies with the highest gross margins in the market. *Portfolio_{LOW}* will include companies with the lowest gross margin in the market. Then, we will compare these portfolios for the current month. We will statistically evaluate if there is any meaningful differences between these two portfolios.

Portfolios

The Stata code to download the most recent annual income statements for all S&P-500 companies:

```
net install http://researchdata.com/stata/048/fetch_statements.pkg, force
net install http://researchdata.com/stata/010/fetchcomponents.pkg, force
fetchcomponents, symbol(~GSPC)
levelsof Symbol, local(symbols) clean
foreach aa in `symbols' {
    di "`aa'"
    clear
    capture: fetch_statements `aa', freq(a) st(IS)
    if (_rc==0) {
        if (_N>5) {
            qui: ds
            local temp = word("`r(varlist)'" ,2)
            rename `temp' most_recent
            keep item most_recent
            gen symbol="`aa'"
            capture: append using SP500_IS.dta
            save SP500_IS.dta, replace
        }
    }
}
```

The Stata code to calculate the gross margins (GM):

```
use SP500_IS.dta, clear
gen temp_rv = most_recent if (item == "Total Revenue")
gen temp_cor = most_recent if (item == "Cost of Revenue")
collapse (firstnm) revenue=temp_rv (firstnm) cost_of_revenue=temp_cor, by(symbol)
gen GM = (revenue - cost_of_revenue) / revenue
sort GM
```

Let's filter our companies that have no cost of revenue (i.e. 1 as a gross profit ratio). Note that most of these companies are financial companies.

```
drop if GM==1
```

The Stata code to get the symbols for companies with the lowest and the highest gross margins. Note that we have 424 companies with usable gross margins.

```
levelsof symbol if _n<=50, local(low_gm) clean
levelsof symbol if _n>(424-50), local(high_gm) clean
```

The Stata code to download the daily prices for the companies with the lowest and the highest gross margins:

```
net install http://researchdata.com/stata/203/fetchyahoquotes.pkg, force
fetchyahoquotes `GSPC' `low_gm' `high_gm', freq(d) chg(ln) start(01jan2018)
```

The Stata code to create the low gross margin and the high gross margin portfolios:

```
gen P_low_gm = 0
foreach aa in `low_gm' {
    replace P_low_gm = P_low_gm + ln`aa' if (ln`aa'!=.)
}
replace P_low_gm = P_low_gm / 50

gen P_high_gm = 0
foreach aa in `high_gm' {
    replace P_high_gm = P_high_gm + ln`aa' if (ln`aa'!=.)
}
replace P_high_gm = P_high_gm / 50
```

Let's compare total risk and return for the recent days (i.e. October and November):

```
tabstat P_low_gm P_high_gm if (month(date)>8), stat(sd sum) columns(variables)
```

stats	P_low_gm	P_high~m
sd	.0110551	.0121923
sum	-.1130482	-.1207151

In this table *sd* refers to standard deviation of daily returns for the two months we are testing. *sum* refers to the total return for the two months. **High gross margin companies had slightly higher risk and had slightly more losses.**

Let's statistically compare daily returns for the recent days (i.e. October and November):

```
ttest P_low_gm == P_high_gm if (month(date)>8)
```

Paired t test

Variable	Obs	Mean	Std. Err.	Std. Dev.	[95% Conf. Interval]
P_low_gm	57	-.0019833	.0014643	.0110551	-.0049166 .00095
P_high~m	57	-.0021178	.0016149	.0121923	-.0053529 .0011172
diff	57	.0001345	.0007806	.0058935	-.0014293 .0016983

```
mean(diff) = mean(P_low_gm - P_high_gm) t = 0.1723
Ho: mean(diff) = 0 degrees of freedom = 56
```

```
Ha: mean(diff) < 0 Ha: mean(diff) != 0 Ha: mean(diff) > 0
Pr(T < t) = 0.5681 Pr(|T| > |t|) = 0.8638 Pr(T > t) = 0.4319
```

Note that this test is based on average daily returns. While both portfolios, high gross margin and

low gross margin, had negative average returns, high gross margin companies had lower returns. The highlighted *t-test* shows that average daily returns are statistically indifferent.

Let's compare AAPL price chart to S&P-500 index for October and November.


Where does AAPL stand?

Let's look at the current distribution of gross margins for the S&P-500 companies:

```
use SP500_IS.dta, clear
gen temp_rv = most_recent if (item == "Total Revenue")
gen temp_cor = most_recent if (item == "Cost of Revenue")
collapse (firstnm) revenue=temp_rv (firstnm) cost_of_revenue=temp_cor, by(symbol)
gen GM = (revenue - cost_of_revenue) / revenue
drop if GM==1
sort GM
generate groupGM=recode(GM,.10,.20,.30,.40,.50,.60,.70,.80,.90)
tabulate groupGM, plot
```


groupGM	Freq.
.1	8 *****
.2	40 *****
.3	45 *****
.4	88 *****
.5	59 *****
.6	45 *****
.7	48 *****
.8	38 *****
.9	53 *****

```
fetchyahoquotes ^GSPC AAPL, freq(d) chg(ln) start(01oct2018)
twoway (line adjclose_AAPL date, yaxis(1)) (line adjclose_GSPC date, yaxis(2))
```



Let's chart the gross margin distribution.

```
rename GM Gross_Margin
hist Gross_Margin, freq bin(20)
```



Note that the gross margin for AAPL is 38.34% as of November 23rd. Current market perception seems not to differentiate companies based on the gross margins.

